

---

Subject: Re: Issues with recalculating wealth index  
Posted by [Trevor-DHS](#) on Tue, 01 Jun 2021 16:02:51 GMT  
[View Forum Message](#) <> [Reply to Message](#)

---

1) The initial extraction of data for the calculation of the wealth index is done in CSPro using the raw data file, not the recode file. Below is the key portion of that code, and following the code a few notes regarding the domestic, house, and land variables.

```
PROC HOUSEHOLD
```

```
preproc
```

```
if AHRESULT <> 1 then
```

```
  skip case;
```

```
endif;
```

```
hhmembers = totocc(AHSEC01_EDT);
```

```
hhusual = count(AHSEC01_EDT where AH05 = 1);
```

```
hhslept = count(AHSEC01_EDT where AH06 = 1);
```

```
domestic = 0;
```

```
land = 0;
```

```
if AH119 = 1 then land = 1; endif;
```

```
house = 0;
```

```
{ if ownership of house asked in household questionnaire !!!}
```

```
{ if AHXXX = 1 then house = 1; endif; }
```

```
postproc
```

```
{ Code for exporting the asset variables - used from postproc of household as some variables are  
created from the individual questionnaires }
```

```
export to wealth_assets case_id (AHCLUST, AHNUMBER)
```

```
  AHWEIGHT, AHREGION, AHTYPE, hhmembers, hhusual, hhslept,
```

```
  AHSEC02 exclude(AH108,AH126),
```

```
  AHSEC04 exclude(AH141),
```

```
  domestic, house, land; { Drop alpha variables !!!}
```

```
frequency include(AHREGION, AHTYPE, hhmembers, hhusual, hhslept,
```

```
  AHSEC02 numeric alpha,
```

```
  AHSEC04 numeric alpha, domestic, house, land)
```

```
exclude(AH108,AH126,AH141); { Drop alpha variables !!!}
```

```
proc INDIVIDUAL
```

```
preproc
```

```
{ Woman is a domestic worker and is not related to head of household }
```

```
if A913 = maid_code and AH03(ALINE) = not_related then domestic = 1; endif;
```

```
{ Woman's partner is a domestic worker, listed in the household and is not related to head of  
household }
```

```
if A908 = maid_code and A705 in 1:AHMEMBER and AH03(A705) = not_related then domestic =  
1; endif;
```

```
{ Man is a domestic worker and is not related to head of household }  
if AM604 = maid_code and AH03(ALINE) = not_related then domestic = 1; endif;
```

```
{ Add code for ownership of land or house if asked in individual women or men's questionnaires -  
!!! }
```

```
{ Only used to set land or house = 1. Already initialized to 0 in the preproc for the household }  
if AH05(ALINE) = 1 & A925 in 1,2,3 then house = 1; endif;  
if AH05(ALINE) = 1 & A928 in 1,2,3 then land = 1; endif;
```

```
if AH05(ALINE) = 1 & AM612 in 1,2,3 then house = 1; endif;
```

```
if AH05(ALINE) = 1 & AM615 in 1,2,3 then land = 1; endif;
```

Notes:

The default code for maid/domestic (although this varies from survey to survey) is code 51. See variables V704, V716, and MV716 for the relevant variables to use. Note also that the person must not be related to the head of the household (HV101(V003) = 12). Also when referring to the husband's occupation in V704, the husband must be listed in the household roster (see V034), and the husband must not be related to the head of the household (HV101(V034) = 12).

For the household and land variables, the respondent to the women's or men's questionnaire must be de jure members of the household (HV102(V003) = 1/HV102(MV003) = 1). The house and land variables may also be set from the household questionnaire. 'Land' is set if HV244 = 1. 'House' may be set if there is a similar variable in the household questionnaire although this is non-standard (in the standard it is only based on the questions in the man's or the woman's questionnaires).

- 2) We will review your notes and check the files available online and get back to you.
- 3) Can you let us know which files are corrupted, and we will check these.